
SPS – Programm 

(Tia Portal / STEP 7 Professional V16) 

Teil 1: Datenbaustein „Variablen“ 

Variable „PID“ (Struct) wird angelegt 

 

 

Teil 2: Organisationbaustein „Cyclic Interrupt“ (Weckalarm). 

Sprache SCL, Zeittakt 5000 ms 

Name: PID_Regler_Berechnungen 

// Stellglieder Heizung und Kühlung AUS --------------------------------------------- 

"Variablen".PID.Heizung := FALSE; 

"Variablen".PID.Kühlung := FALSE; 

 

// Tabelle "Regeldifferenz" neu ordnen ---------------------------------------------- 

"Variablen".PID.Regeldifferenz[3] := "Variablen".PID.Regeldifferenz[2]; 

"Variablen".PID.Regeldifferenz[2] := "Variablen".PID.Regeldifferenz[1]; 

"Variablen".PID.Regeldifferenz[1] := "Variablen".PID.Soll_Temperatur - "Variablen".PID.Ist_Temperatur; 

 

// Berechnungen durchführen --------------------------------------------------------- 

#Wert[1] := "Variablen".PID.Regeldifferenz[1] - "Variablen".PID.Regeldifferenz[2]; 

#Wert[2] := "Variablen".PID.I * "Variablen".PID.Regeldifferenz[1]; 

#Wert[3] := "Variablen".PID.Regeldifferenz[1] - (2 * "Variablen".PID.Regeldifferenz[2]) + 

"Variablen".PID.Regeldifferenz[3]; 

#Wert[3] := "Variablen".PID.D * #Wert[3]; 



"Variablen".PID.Stellwert := "Variablen".PID.P * (#Wert[1] + #Wert[2] + #Wert[3]); 

 

// Entscheidung Heizung oder Kühlung EIN -------------------------------------------- 

IF ("Variablen".PID.Stellwert > 0) THEN 

     "Variablen".PID.Heizung := TRUE; 

END_IF; 

IF ("Variablen".PID.Stellwert < 0) THEN 

    "Variablen".PID.Kühlung := TRUE; 

END_IF; 

 

// Variable Stellenwert bestimmen --------------------------------------------------- 

"Variablen".PID.Stellwert := ABS("Variablen".PID.Stellwert); 

IF "Variablen".PID.Stellwert > "Variablen".PID.Abtastzeit THEN 

    "Variablen".PID.Stellwert := "Variablen".PID.Abtastzeit; 

END_IF; 

 

// Start-Merker für den Timer aktivieren --------------------------------------------- 

"Variablen".PID.Zeit_Start := TRUE; 

 

Teil 3: Funktionsbaustein (oder Funktion) 

Sprache SCL 

Name: PID_Regler 

INPUT 

 

OUTPUT 

 



Variablen 

 

 

// Ausgänge zurücksetzen ---------------------------------------- 

#"Heizung e/a" := FALSE;  

#"Kühlung e/a" := FALSE; 

 

// Variablen für "PID_Regler_Berechnungen" bereitstellen -------- 

"Variablen".PID.Soll_Temperatur := #"SOLL_Temperatur [°C]"; 

"Variablen".PID.Ist_Temperatur := #"IST_Temperatur [°C]"; 

"Variablen".PID.P := #"P - Konstante"; 

"Variablen".PID.I := #"I - Konstante"; 

"Variablen".PID.D := #"D - Konstante"; 

"Variablen".PID.Abtastzeit := #"Abtastzeit [ms]"; 

   

// Stellzeit für den Impuls-Timer bereitstellen ----------------- 

#Stellzeit := REAL_TO_DINT("Variablen".PID.Stellwert) - 10; 

 

// Timer start -------------------------------------------------- 

  "IEC_Timer_0_DB".TP(IN:="Variablen".PID.Zeit_Start, 

                      PT:=#Stellzeit, 

                      Q=>#Out, 

                      ET=>#t_ist);  

   

// Start Merker zurücksetzen ------------------------------------ 

IF NOT #Out THEN 

    "Variablen".PID.Zeit_Start := FALSE; 

END_IF; 

 

// Ausgänge für Heizung und Kühlung aktualisieren --------------- 



IF "PID Start" THEN 

    IF "Variablen".PID.Heizung AND #Out THEN 

        #"Heizung e/a" := TRUE; 

    END_IF; 

    IF "Variablen".PID.Kühlung AND #Out THEN 

        #"Kühlung e/a" := TRUE; 

    END_IF; 

END_IF; 

 

Teil 4: Funktionsbaustein „PID_Regler“ in Hauptprogramm einbinden 

und parametrieren 

Sprache KOP 

(Soll-Temperatur 50°C = 13700 (Peripheriewert) 

 


